NAG Library Routine Document

D01PAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

D01PAF returns a sequence of approximations to the integral of a function over a multi-dimensional simplex, together with an error estimate for the last approximation.

2 Specification

```
SUBROUTINE D01PAF(NDIM, VERT, LDVERT, SDVERT, FUNCTN, MINORD, MAXORD,
1                 FINVLS, ESTERR, IFAIL)
INTEGER           NDIM, LDVERT, SDVERT, MINORD, MAXORD, IFAIL
double precision  VERT(LDVERT,SDVERT), FUNCTN, FINVLS(MAXORD), ESTERR
EXTERNAL          FUNCTN
```

3 Description

D01PAF computes a sequence of approximations FINVLS(j), for $j = \text{MINORD} + 1$, $\text{MINORD} + 2, \ldots, \text{MAXORD}$, to an integral

$$\int_S f(x_1, x_2, \ldots, x_n)\, dx_1 dx_2 \cdots dx_n$$

where S is an n-dimensional simplex defined in terms of its $n + 1$ vertices. FINVLS(j) is an approximation which will be exact (except for rounding errors) whenever the integrand is a polynomial of total degree $2j - 1$ or less.

The type of method used has been described in Grundmann and Moller (1978), and is implemented in an extrapolated form using the theory from de Doncker (1979).

4 References

de Doncker E (1979) New Euler–Maclaurin Expansions and their application to quadrature over the s-dimensional simplex *Math. Comput.* **33** 1003–1018

Grundmann A and Moller H M (1978) Invariant integration formulas for the n-simplex by combinatorial methods *SIAM J. Numer. Anal.* **15** 282–290

5 Parameters

1: NDIM – INTEGER *Input*

On entry: n, the number of dimensions of the integral.

Constraint: NDIM ≥ 2.

2: VERT(LDVERT,SDVERT) – ***double precision*** array *Input/Output*

On entry: VERT(i, j) must be set to the jth component of the ith vertex for the simplex integration region, for $i = 1, 2, \ldots, n + 1$ and $j = 1, 2, \ldots, n$. If MINORD > 0, VERT must be unchanged since the previous call of D01PAF.

On exit: these values are unchanged. The rest of the array VERT is used for workspace and contains information to be used if another call of D01PAF is made with MINORD > 0. In particular VERT$(n + 1, 2n + 2)$ contains the volume of the simplex.

3: LDVERT – INTEGER *Input*

On entry: the first dimension of the array VERT as declared in the (sub)program from which D01PAF is called.

Constraint: LDVERT $\geq$ NDIM $+ 1$.

4: SDVERT – INTEGER *Input*

On entry: the second dimension of the array VERT as declared in the (sub)program from which D01PAF is called.

Constraint: SDVERT $\geq 2 \times$ (NDIM $+ 1$).

5: FUNCTN – ***double precision*** FUNCTION, supplied by the user. *External Procedure*

FUNCTN must return the value of the integrand f at a given point.

The specification of FUNCTN is:

```
double precision FUNCTION FUNCTN(NDIM, X)
INTEGER                   NDIM
double precision          X(NDIM)
```

1: NDIM – INTEGER *Input*

On entry: n, the number of dimensions of the integral.

2: X(NDIM) – ***double precision*** array *Input*

On entry: the co-ordinates of the point at which the integrand f must be evaluated.

FUNCTN must be declared as EXTERNAL in the (sub)program from which D01PAF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

6: MINORD – INTEGER *Input/Output*

On entry: must specify the highest order of the approximations currently available in the array FINVLS. MINORD $= 0$ indicates an initial call; MINORD > 0 indicates that FINVLS(1), FINVLS(2), $\ldots$, FINVLS(MINORD) have already been computed in a previous call of D01PAF.

Constraint: MINORD ≥ 0.

On exit: MINORD $=$ MAXORD.

7: MAXORD – INTEGER *Input*

On entry: the highest order of approximation to the integral to be computed.

Constraint: MAXORD $>$ MINORD.

8: FINVLS(MAXORD) – ***double precision*** array *Input/Output*

On entry: FINVLS(1), FINVLS(2), $\ldots$, FINVLS(MINORD) must contain approximations to the integral previously computed by D01PAF.

On exit: contains these values unchanged, and the newly computed values FINVLS(MINORD $+ 1$), FINVLS(MINORD $+ 2$), $\ldots$, FINVLS(MAXORD). FINVLS(j) is an approximation to the integral of polynomial degree $2j - 1$.

9: ESTERR – ***double precision*** *Output*

On exit: an absolute error estimate for FINVLS(MAXORD).

10: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value −1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value −1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, NDIM < 2,
or LDVERT < NDIM + 1,
or SDVERT < 2 × (NDIM + 1),
or MINORD < 0,
or MAXORD ≤ MINORD.

IFAIL = 2

The volume of the simplex integration region (computed as a determinant by F03AAF) is too large or too small to be representable in the machine.

7 Accuracy

An absolute error estimate is output through the parameter ESTERR.

8 Further Comments

The running time for D01PAF will usually be dominated by the time used to evaluate the integrand FUNCTN. The maximum time that could be used by D01PAF will be approximately given by

$$T \times \frac{(\text{MAXORD} + \text{NDIM})!}{(\text{MAXORD} - 1)!(\text{NDIM} + 1)!}$$

where T is the time needed for one call of FUNCTN.

9 Example

This example demonstrates the use of the subroutine with the integral

$$\int_0^1 \int_0^{1-x} \int_0^{1-x-y} \exp(x+y+z)\cos(x+y+z)\,dz\,dy\,dx = \tfrac{1}{4}.$$

9.1 Program Text

```
*     D01PAF Example Program Text
*     Mark 14 Revised. NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NDIM, LDVERT, SDVERT, MXORD
      PARAMETER        (NDIM=3,LDVERT=NDIM+1,SDVERT=2*(NDIM+1),MXORD=5)
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*     .. Local Scalars ..
```

```
      DOUBLE PRECISION ESTERR
      INTEGER          IFAIL, J, K, MAXORD, MINORD, NEVALS
*     .. Local Arrays ..
      DOUBLE PRECISION FINVLS(MXORD), VERT(LDVERT,SDVERT)
*     .. External Functions ..
      DOUBLE PRECISION FUNCTN
      EXTERNAL         FUNCTN
*     .. External Subroutines ..
      EXTERNAL         D01PAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'D01PAF Example Program Results'
      DO 40 J = 1, LDVERT
         DO 20 K = 1, NDIM
            VERT(J,K) = 0.0D0
   20    CONTINUE
         IF (J.GT.1) VERT(J,J-1) = 1.0D0
   40 CONTINUE
      MINORD = 0
      NEVALS = 1
      WRITE (NOUT,*)
      WRITE (NOUT,*)
     +  'MAXORD   Estimated      Estimated         Integrand'
      WRITE (NOUT,*)
     +  '           value         accuracy        evaluations'
      DO 60 MAXORD = 1, MXORD
         IFAIL = 1
*
         CALL D01PAF(NDIM,VERT,LDVERT,SDVERT,FUNCTN,MINORD,MAXORD,
     +              FINVLS,ESTERR,IFAIL)
*
         IF (IFAIL.EQ.0) THEN
            WRITE (NOUT,99999) MAXORD, FINVLS(MAXORD), ESTERR, NEVALS
            NEVALS = (NEVALS*(MAXORD+NDIM+1))/MAXORD
         ELSE
            WRITE (NOUT,*)
            WRITE (NOUT,99998) ' ** D01PAF returned with IFAIL = ',
     +        IFAIL
            GO TO 80
         END IF
   60 CONTINUE
   80 CONTINUE
*
99999 FORMAT (1X,I4,F13.5,E16.3,I15)
99998 FORMAT (1X,A,I5)
      END
*
      DOUBLE PRECISION FUNCTION FUNCTN(NDIM,X)
*     .. Scalar Arguments ..
      INTEGER          NDIM
*     .. Array Arguments ..
      DOUBLE PRECISION X(NDIM)
*     .. Intrinsic Functions ..
      INTRINSIC        COS, EXP
*     .. Executable Statements ..
      FUNCTN = EXP(X(1)+X(2)+X(3))*COS(X(1)+X(2)+X(3))
      RETURN
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 D01PAF Example Program Results

 MAXORD   Estimated      Estimated         Integrand
            value         accuracy       evaluations
    1      0.25816       0.258E+00             1
    2      0.25011       0.806E-02             5
    3      0.25000       0.107E-03            15
    4      0.25000       0.410E-06            35
    5      0.25000       0.173E-08            70
```